Statistical Physics II.

Landau diamagnetism T=0

De Hass-Van Alphen oscillations

(W. J. De Hass and P. M. Van Halphen)

- **We study a simple model in which electrons move in two dimensions. We have computed the energy per particle and we will plot it. (We follow the classroom discussion)**

  - We compute the first j such that the x values fullfil $\frac{1}{j+1} > x > \frac{1}{j+2}$

```
LogPlot[IntegerPart[1 / x], {x, 2, 0.1},
 PlotStyle → {Thick}, PlotLabel → "In Log scale!",
 Frame → True, FrameLabel → {"x", "IntegerPart[1/x]"}]
```



  - We see the as x decreases the j's for which the inequality studied has solution also increases. j start a 0 when x crosses down 1. When $x = 1/2$ j jumps to 2, etc.

- **Once we have selected for a given x the corresponding j, we assingn the corresponding function**

```
func[j_, x_] := x ((2 j + 3) - (j + 1) (j + 2) x) (* Note that for j=
   -1 we also obtain the case in which x>1 !! *)
```

- Look at the functions and their x-derivatives.

```
functable = Table[func[j, x], {j, -1, 4}]
dfunctable = Table[D[func[j, x], x], {j, -1, 4}]
Table[IntegerPart[1 / x], {x, 0.1, 2, 0.1}]
```

```
{x, (3 - 2 x) x, (5 - 6 x) x, (7 - 12 x) x, (9 - 20 x) x, (11 - 30 x) x}
```

```
{1, 3 - 4 x, 5 - 12 x, 7 - 24 x, 9 - 40 x, 11 - 60 x}
```

```
{10, 5, 3, 2, 2, 1, 1, 1, 1, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

- The Energy function

```
Energy[x_] := func[IntegerPart[1 / x] - 1, x]
```

```
graphenergy = Plot[Tooltip[Energy[x], "Energy"], {x, 0.2, 1.2},
  Frame → True, FrameLabel → {"X=H/H0", "E"/"μBH0"}, PlotStyle → Thick]
```



- The Magnetization function

```
Magnetization[x_] := -D[func[IntegerPart[1 / x] - 1, y], y] /. y -> x
```

```
graphmagnetization =
 Plot[{1, Tooltip[Magnetization[x], "Magnetization"], -1},
  {x, 0.2, 1.2}, PlotRange → All,
  PlotStyle → {Dashed, {Orange, Thick}, Dashed},
  Frame → True, FrameLabel → {"X=H/H0", "𝓜"/"μBn"}, Filling → Axis]
```




```
Show[graphmagnetization, graphenergy, Background → GrayLevel[.9]]
```



- Susceptibility

```
Susceptibility[x_] :=
 -D[func[IntegerPart[1/x] - 1, y], {y, 2}] /. y -> x
```

```
graphsusceptibility = Plot[Susceptibility[x], {x, 0.2, 1.2},
  PlotRange → All, PlotStyle → {Green, Thick}, Frame → True,
  FrameLabel → {"x", "χ H0"/"2μBn"}, Background → GrayLevel[.9]]
```



- Some useful notions, Plot funciton

Plot a function

```
Plot[Sin[x], {x, 0, 10}]
```



- **Plot a list of functions**

```
SineList = Table[Sin[w x], {w, 1, Pi, Pi / 10}]
Plot[SineList, {x, 0, 2 Pi}]
```

$\{\mathrm{Sin}[x], \mathrm{Sin}[(1+\frac{\pi}{10})\, x], \mathrm{Sin}[(1+\frac{\pi}{5})\, x], \mathrm{Sin}[(1+\frac{3\pi}{10})\, x], \mathrm{Sin}[(1+\frac{2\pi}{5})\, x], \mathrm{Sin}[(1+\frac{\pi}{2})\, x], \mathrm{Sin}[(1+\frac{3\pi}{5})\, x]\}$